

Laplace Transforms

Laplace Transform

The definition of the Laplace transform is

$$X[s] = \int_{0^-}^{\infty} x(t) e^{-st} dt$$

The use of 0^- for the left most limit maybe

different from other books. Some use 0^- and some just 0 , but these are all the same if there are no singularities at $t = 0$. When dealing with causal signals, the unilateral and bilateral Laplace transforms are the same. This is because $x(t) = 0$ for $t < 0$.

Mathematica has this function

built in to its library. The syntax is follows :

- `LaplaceTransform[expr, t, s]` gives the Laplace transform of `expr`
- `LaplaceTransform[expr, {t1, t2 ...}, {s1, s2, ..}]` gives the multidimensional Laplace transform of `expr`.

```
Needs["Calculus`LaplaceTransform`"];
```

```
Get::noopen : Cannot open Calculus`LaplaceTransform`. >>
```

```
Needs::nocont :
```

```
Context Calculus`LaplaceTransform` was not created when Needs was evaluated. >>
```


Basic Exercises

Laplace Transforms of Elementary Functions

```
LaplaceTransform[ Sin[t], t,s]
```

$$\frac{1}{1+s^2}$$


```
Plot[ LaplaceTransform[ Sin[t], t,s], {s,0,10}]
```



```
LaplaceTransform[ t^3 Exp[a t], t, s]
```

$$\frac{6}{(a-s)^4}$$

```
Plot[LaplaceTransform[ t^3 Exp[3 t], t,s], {s, 0, 10},  
PlotRange ->{0,100}]
```




Note the Laplace Transform is defined only for $s > 3$, just as described in the theory.

```
LaplaceTransform[ t^Pi, t,s]
```

```
 $s^{-1-\pi} \text{Gamma}[1 + \pi]$ 
```

```
Plot[ LaplaceTransform[ t^Pi, t,s], {s, 0,10}]
```



Make a conjecture about the behavior of a Laplace Transform as the argument $s \rightarrow \text{Infinity}$, based on these graphs.

```
InverseLaplaceTransform[1/(s^2 + 5)^2, s,t]
```

$$\frac{1}{50} \left(-5 t \text{Cos}[\sqrt{5} t] + \sqrt{5} \text{Sin}[\sqrt{5} t] \right)$$

```
Simplify[%]
```

$$\frac{1}{50} \left(-5 t \text{Cos}[\sqrt{5} t] + \sqrt{5} \text{Sin}[\sqrt{5} t] \right)$$

Solving Differential Equations with Laplace Transforms

```
algeqn =
LaplaceTransform[ y''[t] - 2 y'[t] + 5 y[t] == 1 + t, t,s]
```

$$5 \text{LaplaceTransform}[y[t], t, s] + s^2 \text{LaplaceTransform}[y[t], t, s] - 2 (s \text{LaplaceTransform}[y[t], t, s] - y[0]) - s y[0] - y'[0] = \frac{1}{s^2} + \frac{1}{s}$$

```
algeqn = algeqn /. {y[0] -> 0, y'[0] -> 4}
```

$$-4 + 5 \text{LaplaceTransform}[y[t], t, s] - 2 s \text{LaplaceTransform}[y[t], t, s] + s^2 \text{LaplaceTransform}[y[t], t, s] = \frac{1}{s^2} + \frac{1}{s}$$

```
algsoln = Solve[ %, LaplaceTransform[ y[t], t,s] ]
```

$$\left\{ \left\{ \text{LaplaceTransform}[y[t], t, s] \rightarrow \frac{1 + s + 4 s^2}{s^2 (5 - 2 s + s^2)} \right\} \right\}$$

```
Y = LaplaceTransform[ y[t], t,s] /. %
```

$$\left\{ \frac{1 + s + 4 s^2}{s^2 (5 - 2 s + s^2)} \right\}$$

```
Y = First[Y]
```

$$\frac{1 + s + 4 s^2}{s^2 (5 - 2 s + s^2)}$$

```
Apart[ Y,s]
```

$$\frac{1}{5 s^2} + \frac{7}{25 s} + \frac{109 - 7 s}{25 (5 - 2 s + s^2)}$$

```
InverseLaplaceTransform[%, s, t]
```

$$\frac{7}{25} - \left(\frac{1}{50} + \frac{i}{50} \right) e^{(1-2i)t} \left((-22 - 29i) + (29 + 22i) e^{4it} \right) + \frac{t}{5}$$

```
Simplify[%]
```

$$\frac{7}{25} - \left(\frac{7}{50} - \frac{51i}{50} \right) e^{(1-2i)t} - \left(\frac{7}{50} + \frac{51i}{50} \right) e^{(1+2i)t} + \frac{t}{5}$$

Laplace Transforms of Step and Periodic Functions


What Can *Mathematica* Do?

■ Warm-ups

```
InverseLaplaceTransform[ Exp[-4s]/(s^2 + 4), s, t]
```

$$\frac{1}{2} \text{HeavisideTheta}[-4 + t] \text{Sin}[2(-4 + t)]$$

```
Plot[%, {t,0,10}]
```




■ A Diff Eq with a Step Function Right-Hand Side

```
NDSolve[ {y'[t] + y[t] == If[t <= 4, 1, -1], y[0] == 0},  
y, {t,0,10},  
MaxSteps->1000]
```


```
{y -> InterpolatingFunction[{{0., 10.}}, <>]}
```

```
Plot[Evaluate[y[t]/. %], {t,0,10},  
PlotRange-> All]
```



■ A Diff Eq with a Periodic Right-Hand Side


```
Plot[Which[t<1, 1, t<2, -1, t<3, 1, t<4, -1, t<=5, 1],  
{t,0,5}]
```




```
NDSolve[ {y'[t] + y[t] ==  
  Which[t<1, 1, t<2, -1, t<3, 1, t<4, -1, t<=5, 1],  
  y[0] == 0},  
  y, {t,0,5}]
```

```
{y -> InterpolatingFunction[{{0., 5.}}, <>]}
```


```
Plot[ Evaluate[y[t] /. %], {t,0,5}]
```



What *Mathematica* Can't Do (Directly)

```
HeavisideStep[a_,x_] := If[ x>=a, 1,0]
```

```
Plot[ HeavisideStep[1,x], {x,0,3}]
```



```
LaplaceTransform[HeavisideStep[1,t], t,s]
```

$$\frac{e^{-s}}{s}$$


On the other hand, using the proper *Mathematica* definitions.....

```
LaplaceTransform[ UnitStep[t-1], t,s]
```

$$\frac{e^{-s}}{s}$$

```
SquareWave[x_] := If[EvenQ[Floor[x]], 1, -1]
```

```
Plot[ SquareWave[t], {t,0,10}]
```




```
LaplaceTransform[ Abs[ Sin[t]], t,s]
```

$$\frac{1 + e^{\pi s}}{(-1 + e^{\pi s}) (1 + s^2)}$$

Dirac Delta Functions

```
Plot[ DiracDelta[t-1], {t,-1,1}]
```



```
Integrate[ DiracDelta[t-1] Exp[t], {t,0,2}]
```

e


```
LaplaceTransform[ DiracDelta[t-1] Exp[t], t,s]
```

e^{1-s}

```
LaplaceTransform[ DiracDelta[t-1], t,s]
```

e^{-s}

```
Plot[ %, {s,0,10}]
```



```
LaplaceTransform[ DiracDelta[t], t,s]
```

```
1
```

What does this say your previous conjecture about the limiting behavior of Laplace transforms?

Let's Solve

$$y'' + 2 y' + 4 y = t \exp(-t)$$

$$y(0) = 1$$

$$y'(0) = -1$$

```
lhsone = LaplaceTransform[y''[t] + 2 y'[t] + 4 y[t], t, s]
```

```
4 LaplaceTransform[y[t], t, s] + s^2 LaplaceTransform[y[t], t, s] +
  2 (s LaplaceTransform[y[t], t, s] - y[0]) - s y[0] - y'[0]
```

After taking the Laplace Transform of the differential equation, we can plug

in the initial values and make a substitution for the Transform of y (usually

denoted capital Y , which, remember, is still unknown); we do this with the following command :

```
lhstwo =
  lhsone /. {LaplaceTransform[y[t], t, s] → capy, y[0] → 1, y'[0] → -1}
```

```
1 + 4 capy - s + capy s2 + 2 (-1 + capy s)
```

Now we compute the transform of the right hand side:

```
rhs = LaplaceTransform[t - Exp[-t], t, s]
```

$$\frac{1}{s^2} - \frac{1}{1+s}$$

and solve for capital y .

```
stepthree = Solve[lhstwo == rhs, capy]
```

$$\left\{ \left\{ \text{capy} \rightarrow \frac{1 + s + 2 s^3 + s^4}{s^2 (1 + s) (4 + 2 s + s^2)} \right\} \right\}$$

Now we must extract the solution from all of the surrounding brackets with the following command:

```
stepfour = stepthree[[1, 1, 2]]
```

$$\frac{1 + s + 2 s^3 + s^4}{s^2 (1 + s) (4 + 2 s + s^2)}$$

so we can then compute the inverse transform and find the solution to our problem!

To avoid having to retype InverseLaplaceTransform every time you

want to perform the inverse transform I suggest creating an alias such as ILT by simply performing the assignment

```
ILT = InverseLaplaceTransform
```

```
InverseLaplaceTransform
```

within Mathematica. Now we can proceed as before but without having to do as \much typing. Nice!

```
sol = ILT[stepfour, s, t]
```

$$\frac{1}{24} e^{-t} \left(-8 - 3 e^t + 6 e^t t + 35 \cos[\sqrt{3} t] - \sqrt{3} \sin[\sqrt{3} t] \right)$$

The above six steps can be used as a general recipe for solving differential \equations using the Laplace Transform provided Mathematica knows the transform of the left and right hand side. In some cases you may have to do some of the work on your own "by hand" before proceeding.

Finally!

```
lhsone = LaplaceTransform[y''[t] + 2 y'[t] + 4 y[t], t, s]
lhstwo =
  lhsone /. {LaplaceTransform[y[t], t, s] -> capy, y[0] -> 1, y'[0] -> -1}
rhs = LaplaceTransform[UnitStep[t - 5], t, s]
stepthree = Solve[lhstwo == rhs, capy]
stepfour = stepthree[[1, 1, 2]]
sol = ILT[stepfour, s, t]
Plot[sol, {t, -1, 15}]
```

```
4 LaplaceTransform[y[t], t, s] + s^2 LaplaceTransform[y[t], t, s] +
  2 (s LaplaceTransform[y[t], t, s] - y[0]) - s y[0] - y'[0]
```


```
1 + 4 capy - s + capy s^2 + 2 (-1 + capy s)
```

$$\frac{e^{-5s}}{s}$$

$$\left\{ \text{copy} \rightarrow \frac{e^{-5s} (1 + e^{5s} s + e^{5s} s^2)}{s (4 + 2s + s^2)} \right\}$$

$$\frac{e^{-5s} (1 + e^{5s} s + e^{5s} s^2)}{s (4 + 2s + s^2)}$$

$$\frac{1}{12} e^{-t} \left(12 \text{Cos}[\sqrt{3} t] + \text{HeavisideTheta}[-5 + t] \right. \\ \left. \left(3 e^t - 3 e^5 \text{Cos}[\sqrt{3} (-5 + t)] - \sqrt{3} e^5 \text{Sin}[\sqrt{3} (-5 + t)] \right) \right)$$



Exercise 1

Solve

$$y'' + 5y' - y = \exp(t) - 1$$

$$y(0) = 1$$

$$y'(0) = 1$$